
Documentation

PJON is designed to be as user friendly and minimally technical as possible, so it can be quickly mastered by new users. Visit the dedicated sections to get detailed info:



	Addressing [https://github.com/gioblu/PJON/tree/master/documentation/addressing.md]

	Configuration [https://github.com/gioblu/PJON/tree/master/documentation/configuration.md]

	Data reception [https://github.com/gioblu/PJON/tree/master/documentation/data-reception.md]

	Data transmission [https://github.com/gioblu/PJON/tree/master/documentation/data-transmission.md]

	Error handling [https://github.com/gioblu/PJON/tree/master/documentation/error-handling.md]

	IO setup [https://github.com/gioblu/PJON/tree/master/documentation/io-setup.md]





[image: Join the chat at https://gitter.im/gioblu/PJON] [https://gitter.im/gioblu/PJON?utm_source=badge&utm_medium=badge&utm_campaign=pr-badge&utm_content=badge] [image: Donate] [https://www.paypal.me/PJON]

Feel free to write me personally gioscarab@gmail.com





          

      

      

    

  

    
      
          
            
  
Interfacing

PJON is designed to be as user friendly and less technical as possible to be fastly mastered also by new users.
Visit the architecture dedicated sections to get detailed info on your setup:



	ATtiny interfacing

	ESP8266 interfacing

	Deal with interference

	Troubleshooting





[image: Join the chat at https://gitter.im/gioblu/PJON] [https://gitter.im/gioblu/PJON?utm_source=badge&utm_medium=badge&utm_campaign=pr-badge&utm_content=badge] [image: Donate] [https://www.paypal.me/PJON]

Feel free to write me personally gioscarab@gmail.com





          

      

      

    

  

    
      
          
            
  
ATtiny Interfacing

AVR ATtiny microcontroller family is a really interesting and compact platform supported by PJON Arduino compatible implementation [https://github.com/gioblu/PJON] and soon also by PJON_ASK wireless implementation [https://github.com/gioblu/PJON_ASK].

####How to program ATtiny 45/85
You physically need at least one ATtiny microcontroller, a breadboard, some jumpers and an Arduino duemilanove / Uno used as an Arduino ISP programmer. Follow High-Low Tech tutorial [http://highlowtech.org/?p=1706] by David Mellis and get the last version of the attiny repository [https://github.com/damellis/attiny].

####Use PJON with ATtiny45/85
PJON Arduino compatible implementation [https://github.com/gioblu/PJON] works smoothly on ATtiny45/85 8Mhz (internal) with pin 2 and 3.

####Use PJON with ATtiny45/85 with external oscillator
Because of the internal clock’s lack of precision, with some ATtiny85 in particular, low communication performance can be detected; extended tests proven the ATtiny internal clock to be extremely inaccurate (timing inconsistency between two identical ATtiny85 can be detected). Here is an example how it works with external 16 MHz oscillator.

[image: image]

This is the sketch for Arduino UNO:

#include <PJON.h>
PJON<SoftwareBitBang> bus(1); // <Strategy name> bus(selected device id)
#define LED 13
void setup() {
  pinModeFast(LED, OUTPUT);
  digitalWriteFast(LED, LOW);
  bus.strategy.set_pin(12);
  bus.include_sender_info(false);
  bus.begin();
}
void loop() {  
  digitalWrite(LED, HIGH);
  delay(30);
  digitalWrite(LED, LOW);
  bus.send_packet_blocking(2, "B",1);
  delay(30);
};





This is the sketch for ATtiny85:

#include <PJON.h>
PJON<SoftwareBitBang> bus(2); // <Strategy name> bus(selected device id)
#define LED 0
void setup() {
  pinModeFast(LED, OUTPUT);
  digitalWriteFast(LED, LOW);
  bus.strategy.set_pin(2);
  bus.include_sender_info(false);
  bus.begin();
  bus.set_receiver(receiver_function);
};
void receiver_function(uint8_t *payload, uint8_t length, const PacketInfo &packet_info) {
  digitalWrite(LED, HIGH);
  delay(30);
  digitalWrite(LED, LOW);
}
void loop() {
  bus.receive(1000);
};









          

      

      

    

  

    
      
          
            
  
ESP8266 Interfacing

If you are creating a 3.3v only PJON bus the wiring needed is really simple:

[image: Wiring Diagram]

If you are connecting an ESP8266 to a 5v bus you need to add some complexity to the standard PJON wiring. Logic level converter usage is strongly suggested otherwise you risk to fry ESP8266 pins and to have communication problems due to logic 1 treshold of 5v boards. Here you can find a fritzing project made by aperepel: http://fritzing.org/projects/pjon-with-uno-and-esp8266nodemcu

Note:


	Add 10KOhm pulldown resistors on Arduino Duemilanove / Uno / Nano side.

	Add 10KOhm pulldown resistors on ESP side (converter without integrated pullup resistors).

	Add 1-2KOhm  pulldown resistors on ESP side (converter with integrated pullup resistors).



Here’s the diagram from the project.
[image: Wiring Diagram]





          

      

      

    

  

    
      
          
            
  
Deal with interference

In some situations, interference can affect the communication medium; understanding its source and how to mitigate its effects is a key element to master understanding in electronics and digital communication. Interference affecting a PJON bus can be divided in three different groups:


	Generated by hardware connected to the bus.

	Generated by third-party hardware not connected to the bus or device’s power supply.

	Unknown.



The third category describes interference that can occur because of unknown sources, the extremely complicated and interrelated behavior of radiation waves often brings impossible to determine the cause (or the sum of causes) of a disturbing signal.


Connected hardware related interference

The most experienced sort of interference is a high chance to find the channel BUSY. This happens because the pin is “floating” from a logic state to the other one. This often happens because the bus, and the devices connected to it are acting as a capacitor, often reaching the HIGH logical state also in a quiescent state. A fast solution to this problem is the use of a pull-down resistor to discharge the unwanted capacitance build-up in the channel. The correct resistor value to be applied is very variable, depending on the topology and medium used, but is often near to the mega ohm order.

I/O PIN ------------------- BUS
                      |
GND -------[/\/]------/

           resistor





Bandwidth loss related to the system’s power consumption is the sign of power supply interference. Never feed devices with shared power supply with high-power demanding appliances like servos, motors or actuators. If this is not possible at least position a diode connected to VIN, able to block inverse current from the VIN pin back to the wire and use a capacitor to fill temporary feeding shortages.

GND ------
         |    
        === capacitor
         |
VIN ---------[|__]------- POWER SUPPLY FEED

             diode





The above solution can reduce but not eliminate transient voltage spikes. A more decisive way to deal with transient voltage is to use diodes in opposition. This technique reduces the short duration electrical spikes can occur because of power shortages, power transitions in other large equipment on the same power line or lightning strikes.

             diode                       diode

GND ---------[__|]--------------|--------[__|]--- POWER SUPPLY FEED
                                |                
                /-----[/\/]-----/  
                |
                |     resistor
I/O PIN  -------|-------------------------------- BUS                         





A serious step toward reliability is to use a well insulated wire (mil standard for radio communication / avionics surplus), but this can only be applied for home-made / non-serial production; a more standard approach is the use of a simple coaxial cable in one of its forms, from earbuds wire to super-expensive gold-plated coaxial. Thanks to the “sock” ground shielding the use of this sort of cable highly reduces interference and is also really comfortable, connecting also the ground with only one wire.

 _______                                   _______
|       | I/O PIN                 I/O PIN |       |
|       |\                               /|       |
| ID 1  | > ====== COAXIAL CABLE ====== < | ID 2  |
|       |/                               \|       |
|_______| GND                         GND |_______|








Third-party hardware related interference

Electromagnetic fields can temporary charge the bus and provoke series of burst-errors. This is often provoked by powerful rotating, magnetic motors, welders, tasers and other devices able to burst a mix of high-power radio waves and magnetic fields. A strong palliative is the use of ferrite beads. Many devices where digital communication is used are equipped with it (see Sony Playstation old wired controller).

              _______
             |       |
I/O PIN -----|-------|----- BUS
             |_______|

              ferrite bead











          

      

      

    

  

    
      
          
            
  
Troubleshooting

Also if PJON is designed to be a really stable, interference and error aware communication Standard, noise can variate enormously because of the environment, the setup and medium you are using as communication channel. PJON single bidirectional medium is not a balanced pair, for this reason long distances and interference sources can affect communication reliability and data throughput.

If you have a Saleae Logic Analyzer you will be amazed by saleae-pjon-protocol-analyzer [https://github.com/aperepel/saleae-pjon-protocol-analyzer] crafted by the user aperepel, it offers a complete analysis suite aware of the Standard functional procedures and symbols.

Pull from PJON’s desired implementation repository the master and run the example NetworkAnalysis, SpeedTest and ErrorTest. These 3 sketches are designed to execute a test and respond through the Serial monitor with a benchmark on communication channel performance and reliability:


	Absolute communication speed

	Practical bandwidth or channel throughput

	Number of packets sent in the test window

	How many errors detected with CRC

	How many times the receive function ended with no reception

	How many times the channel is found busy

	Accuracy ( packets sent / packets received with mistakes ratio)



If you detect absent or slow communication speed, a lot of CRC detected mistakes and / or channel often busy, here you can find a list of really common issues that can lead to this problem:


	If necessary (i.e. wire / conductive medium) use common ground for every device.

	Pin configuration in your code.

	Physical wiring to the pin.

	Device ID configuration in your code.

	Other tasks are occupying all the available loop time.

	Uncorrect packet length passed to send() function

	Forgot the update() or receive() function in loop ;)



If you haven’t identified the problem or you are a more advanced user, porting a new device / architecture, here you can find a list of common causes of this problem:

####Range


	Long distance between devices.

	Many failed receptions.

	Many mistakes detected by CRC.



You are probably near the maximum distance range of your system. The most straight-forward solution is to higher transmission power or to use a small capacitor to filter 1s received as 0s because of distance weakened voltage.

####Interference


	Channel detected busy many times.

	Many mistakes detected by CRC.

	Low or absent communication speed.



Device avoids to transmit over noise to ensure correct communication, when the medium is affected by noise, data throughput and communication reliability drops. Because of interference are also detected mistakes by CRC. See the dedicated wiki page Deal with interference.

####Timing


	Many failed receptions.

	Many mistakes detected by CRC.

	Low or absent communication speed.



Bad syncronization or timing configuration. If you are porting a new device or architecture try to tweak BIT_WIDTH, BIT_SPACER, READ_DELAY and ACCEPTANCE in PJON.h and consider that every architecture will execute code with a different timing.

####Execution time


	Low quality of communication also after tuning / timing tweek.

	Many failed receptions.

	Many CRC detected mistakes.

	Low or absent communication speed.



Every architecture needs a different time to execute the same PJON’s code, so at the point where it should start to read the first bit of a byte, after initial padding bits, it can be a little shifted in time relatively to the transmitter, and so not able to read correctly the bit sequence. READ_DELAY constant in PJON.h regulates the correct positioning in every bit of the 8 readings in time. Take in consideration that a still not implemented architecture / device may not be fast enough to run PJON, try using a faster clock or optimize digital I/O (see digitalWriteFast.h).





          

      

      

    

  

    
      
          
            
  
Strategies

A Strategy its a class containing a set of functions able to physically communicate data through the medium used, abstracting the physical layer from PJON procedure and codebase.

boolean can_start(uint8_t input_pin, uint8_t output_pin)





Should Return true if the medium is free for use and false if the medium is in use by some other device.

void send_byte(uint8_t input_pin, uint8_t output_pin)





Sends a byte on a pin

uint16_t receive_byte(uint8_t input_pin, uint8_t output_pin)





Receives a byte from a pin

void send_response(uint8_t response, uint8_t input_pin, uint8_t output_pin)





Send a response to the packet’s transmitter

uint16_t receive_response(uint8_t input_pin, uint8_t output_pin)





Receives a response from the packet’s receiver

You can define your own set of functions to use PJON with your personal strategy on the medium you prefer. As can see two pins are passed to the methods, enabling twisted pair, serial or radio transceiver physical layer strategies. If you need other functions, variables or constants, those can be defined in your personal Strategy class. Other communication protocols could be used inside those functions to transmit data.

// Simple Serial physical layer example
static inline __attribute__((always_inline))
void send_byte(uint8_t b, uint8_t input_pin, uint8_t output_pin) {
  Serial.print(b);
}





####How to define a new strategy
To define your new strategy you have only to create a new folder named for example YourStrategyName in strategies
directory and write the necessary file YourStrategyName.h:

class YourStrategyName {
  public:
    boolean can_start() { ... };
    uint16_t receive_byte(uint8_t input_pin, uint8_t output_pin) { ... };
    void send_byte(uint8_t b, uint8_t input_pin, uint8_t output_pin) { ... };
    uint16_t receive_response(uint8_t input_pin, uint8_t output_pin) { ... };
    void send_response(uint8_t response, input_pin, uint8_t output_pin) { ... };
}





Simply add your code in the functions declaration shown above and instantiate PJON using the strategy type you
have created: PJON<YourStrategyName> bus();.





          

      

      

    

  

    
      
          
            
  
Press

PJON has been released in the public domain through the Arduino forum and www.gioblu.com [http://www.gioblu.com] in the far 2010. Only in 2015, thanks to the first publication of Michael Teew and then soon Hackernews, the PJON’s repository started receiving a lot of traffic and new users. Since that day PJON has been reviewed by some of the most visited websites and blogs:



	Hackaday blog [http://hackaday.com/2016/03/31/pjon-fancy-one-wire-arduino-communications-protocol-for-home-automation/]

	ATMEL blog [http://blog.atmel.com/2016/04/09/pjon-is-a-pretty-cool-one-wire-protocol/?utm_source=rss&utm_medium=rss]

	Hackernews [https://news.ycombinator.com/item?id=10020625]

	Ardumania [http://www.ardumania.es/pjon-one-wire-introduccion/]

	Michael Teew’s blog [http://michaelteeuw.nl/post/130558526217/pjon-my-son]

	Studio Pieters blog [http://www.studiopieters.nl/communicate-pjon/]

IO Pins Steup

If you want to communicate bidirectionally on a single medium, sharing transmission and reception line, use:

 PJON<> bus;
 bus.strategy.set_pin(12);





With this setup is possible to communicate in HALF_DUPLEX with up to 254 different devices on the same wire.

If for some reason you need to keep separate the two lines, for example if using cheap radio transmitter and receiver modules, to which you have to connect two wires:

 PJON<OverSampling> bus;
 bus.strategy.set_pins(11, 12);





If you don’t need bidirectional communication and you have only the transmitter on one side and the receiver on the other side you can use the NOT_ASSIGNED constant:

 PJON<OverSampling> bus;
 bus.strategy.set_pins(11, NOT_ASSIGNED); // Only receiver









          

      

      

    

  

    
      
          
            
  
Software BitBang

Medium: Wire |
Pins used: 1 / 2

SoftwareBitBang is the default physical layer strategy used by the PJON template object. This implementation is based on micros() and delayMicroseconds(). It makes no use of dedicated timers or interrupt driven strategies to handle binary communication. It is designed to have a small footprint on memory and to be extremely resilient to interference and timing inaccuracies. Thanks to the use of a dedicated digitalWriteFast library, can be achieved fast and reliable cross-architecture communication through one or two pins.

####Byte transmission
Every byte is prepended with 2 synchronization padding bits and transmission occurs LSB-first. The first is a longer than standard logic 1 followed by a standard logic 0. The reception tecnique is based on finding a logic 1 as long as the first padding bit within a certain threshold, synchronizing to its falling edge and checking if it is followed by a logic 0. If this pattern is recognised, reception starts, if not, interference, synchronization loss or simply absence of communication is detected at byte level.

 __________ ___________________________
| SyncPad  | Byte                      |
|______    |___       ___     _____    |
|  |   |   |   |     |   |   |     |   |
|  | 1 | 0 | 1 | 0 0 | 1 | 0 | 1 1 | 0 |
|_ |___|___|___|_____|___|___|_____|___|
   |
 ACCEPTANCE





This adds a certain overhead to information but reduces the need of precise time tuning because synchronization is renewed every byte. All the first padding bit duration minus ACCEPTANCE is the synchronization window the receiver has for every incoming byte. If the length of the first padding bit is less than ACCEPTANCE the received signal is considered interference.

####How to use SoftwareBitBang
Pass the SoftwareBitBang type as PJON template parameter to instantiate a PJON object ready to communicate in this Strategy. All the other necessary information is present in the general Documentation.

  PJON<SoftwareBitBang> bus;





####Compatibility


	ATmega88/168/328 16Mhz (Diecimila, Duemilanove, Uno, Nano, Mini, Lillypad)

	ATmega2560 16Mhz (Arduino Mega)

	ATmega16u4/32u4 16Mhz (Arduino Leonardo)

	ATtiny45/85 8Mhz, see ATtiny Interfacing

	SAMD (Arduino Zero)

	ESP8266 v.1-7 80Mhz “AI-THINKER AT” firmware, see https://github.com/esp8266/Arduino

	ESP8266 NodeMCU v0.9-1.0 80Mhz, see https://github.com/esp8266/Arduino

	MK20DX256 96Mhz (Teensy 3.1)



####Performance
PJON works in 3 different communication modes, STANDARD, FAST and OVERDRIVE:


	STANDARD runs at 16944Bd or 2.12kB/s cross-architecture, promiscuous clock compatible.

	FAST runs at 25157Bd or 3.15kB/s cross-architecture, promiscuous clock compatible.

	OVERDRIVE runs a specific architecture at its maximum limits (non cross-architecture compatible). Every architecture has its own limits, Arduino Duemilanove for example runs at 33898Bd or 4.23kB/s, Arduino Zero can reach 48000Bd or 6.00kB/s.



When including and using SoftwareBitBang, as physical layer of a PJON bus, you have the complete access to the microntroller ready to be used, as usual, untouched. This happens because SoftwareBitBang is completely software emulated strategy with a non blocking implementation, transforming a painfull walk to the hill in a nice flight.

Single wire simplicity let you to experiment quickly and with creativity. The first test I suggest, at your own risk, is to let two arduino boards communicate through your body touching with the left hand the digital port of the first board (5v 40ma, harmless) and with the right the port of the other one. Will be stunning to see high accuracy digital communication running inside a living biological body. This opens the mind to possible creative solutions.

####Why not interrupts?
Usage of libraries is really extensive in the Arduino environment and often the end user is not able to go over collisions or redefinitions. Very often a library is using hardware resources of the microcontroller as timers or interrupts, colliding or interrupting other libraries. This happens because in general Arduino boards have limited hardware resources. To have a universal and reliable communication medium in this sort of environment, software emulated bit-banging, is a good, stable and reliable solution that leads to “more predictable” results than interrupt driven systems coexisting on small microcontrollers without the original developer and the end user knowing about it.

[image: PJON - Michael Teeuw application example]

PJON application example made by the user Michael Teeuw [http://michaelteeuw.nl/post/130558526217/pjon-my-son]





          

      

      

    

  

    
      
          
            
  
Add Configuration

Before approaching to the PJON class it is possible to define the packets and content buffer length.  Pre-defining MAX_PACKETS and PACKET_MAX_LENGTH it is possible to configure this constants to reach the project and memory requirements. Obviously, the less memory is dedicated to this buffers, the more memory can be used for something else.

#define MAX_PACKETS 1
#define PACKET_MAX_LENGTH 20
#include <PJON.h>
/* PJON can store up to 1 packet of up to 
   20 characters - packet overhead (from 4 to 13 depending by configuration) */





Templates can be scary at first sight, but they are quite straight-forward and efficient. Lets start coding, looking how to instantiate in the simplest way the PJON object that in the example is called bus with a wire compatible physical layer:

  PJON<> bus;

  // or 

  PJON<SoftwareBitBang> bus;





The PJON bus runs by default through the SoftwareBitBang strategy.

If you need to predefine the device’s id on instantiation, pass the selected id as shown below:

  PJON<> bus(1); // Device id 1





If you need to communicate on a noisy medium like radio, laser or infrared light, OverSampling strategy can be used effectively (i.e. using cheap 315/433Mhz radio transcerivers maximum range is up to 5km with line of sight):

  PJON<OverSampling> bus;





Configure network state (local or shared). If local, so if passing false, the PJON protol layer procedure is based on a single byte device id to univocally communicate with a device; if in shared mode, so passing true, the protocol adopts a 4 byte bus id to univocally communicate with a device in a certain bus:

  bus.set_shared_network(true);





If you are transmitting on a shared medium it is better to define a bus id to isolate your networking from other buses nearby and to avoid unwanted and potentially critical collisions (i.e. opening the neighbour’s garage door, when you just wanted to brighten your kitchen):

  // Bus id definition
  uint8_t bus_id[] = {0, 0, 0, 1};

  // PJON object
  PJON<OverSampling> bus(bus_id, 45);





In this example we are defining a new bus id made by 4 bytes, and we are passing it to the PJON object followed by the chosen and predefined device id.

Configure the communication mode:

  bus.set_communication_mode(SIMPLEX);     // Run in mono-directional SIMPLEX mode
  bus.set_communication_mode(HALF_DUPLEX); // Run in bi-directional HALF_DUPLEX mode





Configure acknowledge:

  bus.set_acknowledge(false);





PJON by default includes the sender information in the packet. If you don’t need this information you can use the provided setter to reduce overhead and higher communication speed:

  bus.include_sender_info(false);





Configure your device to act as a router, so receiving all the incoming packets:

  bus.set_router(true);





Avoid packet auto-deletion:

  bus.set_packet_auto_deletion(false);









          

      

      

    

  

    
      
          
            
  
Through Hardware Serial

Medium: Hardware Serial port |
Pins used: 2

With ThroughHardwareSerial PJON can run through the Serial port of your device. Both ports should be free from USB computer connection and data pins should be wired inverted, TX to RX and RX to TX.

####Why PJON over Serial?
Serial communication is an hardware integrated communication medium that can reach very fast communication speed but it is quite basic and lacks many of the necessary structures and entities for an easy setup of a network communication.

####Serial Downsides:


	No CRC, checksum or parity bit (ensure correct data transfert)

	No transmission certainty (lack of acknowledge from receiver)

	No packet handling (no retrasmission in case of error)

	No idea of id (no chance to connect more than 2 devices together)



####Adding PJON over Serial you get:


	Correct transmission certainty with the use of CRC 8-bit

	Acnowledge byte to inform transmitter of correct data reception

	Packet handling, i.e. schedule a repeated packet transmission.

	ThroughHardwareSerial can run a 2 device bus in multi-master mode or a n devices bus in master-slave mode. (n still unkown, tested with 1 master and 2 slaves)



####How to use ThroughHardwareSerial
Pass the ThroughHardwareSerial type as PJON template parameter to instantiate a PJON object ready to communicate in this Strategy.

  PJON<ThroughHardwareSerial> bus; // 2 pin over-sampled physical layer





Call the begin method on the Serial object you want to use for PJON communication and pass it to the set_serial method:

  void setup() {
    Serial.begin(9600);
    bus.strategy.set_serial(&Serial);
  }





All the other necessary information is present in the general Documentation.





          

      

      

    

  

    
      
          
            
  
Trasmit Data

The begin function has to be called in the setup or in the loop. The lack of this method call can lead to collision problems on startup, so be sure to call this function before starting transmitting or receiving.

  bus.begin();





Data transmission is handled by a packet manager, call the update() function once per loop cycle:

  bus.update();





To send a string to another device connected to the bus simply call send() function passing the id you want to contact, the string you want to send and its length:

bus.send(100, "Ciao, this is a test!", 21);





I know that the packet length is boring to fill but is there to prevent buffer overflow. If sending arbitrary values NULL terminator strategy based on strlen() is not safe to detect the end of a string.

To send a value repeatedly simply call send_repeatedly() and add as last parameter the interval in microseconds you want between every sending:

int one_second_test = bus.send_repeatedly(100, "Test sent every second!", 23, 1000000);





The one_second_test variable contains the id of the packet, to remove this repeated task simply:

bus.remove(one_second_test);





To broadcast a message to all connected devices, use the BROADCAST constant as recipient ID.

int broadcastTest = bus.send(BROADCAST, "Message for all connected devices.", 34);









          

      

      

    

  

    
      
          
            
  #Compliant Tools

Here a list of the compliant tools:



	ModuleInterface [https://github.com/fredilarsen/ModuleInterface]

	PJON-python [https://github.com/Girgitt/PJON-python]

	saleae-pjon-protocol-analyzer [https://github.com/aperepel/saleae-pjon-protocol-analyzer]





[image: Join the chat at https://gitter.im/gioblu/PJON] [https://gitter.im/gioblu/PJON?utm_source=badge&utm_medium=badge&utm_campaign=pr-badge&utm_content=badge] [image: Donate] [https://www.paypal.me/PJON]

Feel free to write me personally gioscarab@gmail.com



          

      

      

    

  

    
      
          
            
  
Receive Data

Define a void function that will be called if a correct message is received. This function receives 3 parameters: the transmission content, its length and a pointer to a PacketInfo data structure that contains all the info contained in the packet metadata:

void receiver_function(uint8_t *payload, uint8_t length, const PacketInfo &packet_info) {
  Serial.print("Header: ");
  Serial.print(packet_info.header, BIN);
  // If packet formatted for a shared medium
  if((packet_info.header & MODE_BIT) != 0) {
    Serial.print(" Receiver bus id: ");
    Serial.print(packet_info.receiver_bus_id[0]);
    Serial.print(packet_info.receiver_bus_id[1]);
    Serial.print(packet_info.receiver_bus_id[2]);
    Serial.print(packet_info.receiver_bus_id[3]);
    Serial.print(" Device id: ");
    Serial.print(packet_info.receiver_id);
    // If sender info is included
    if((packet_info.header & SENDER_INFO_BIT) != 0) {
      Serial.print(" Sender bus id: ");
      Serial.print(packet_info.sender_bus_id[0]);
      Serial.print(packet_info.sender_bus_id[1]);
      Serial.print(packet_info.sender_bus_id[2]);
      Serial.print(packet_info.sender_bus_id[3]);
      Serial.print(" device id: ");
      Serial.println(packet_info.sender_id);
    }  
  }

  Serial.print(" Content: ");
  for(uint8_t i = 0; i < length; i++) 
    Serial.print((char)payload[i]);

  Serial.print(" Length: ");
  Serial.println(length);
};





Inform the bus to call receiver_function when a correct message is received:

bus.set_receiver(receiver_function);





To correctly receive data call the receive function at least once per loop cycle:

int response = bus.receive();





If you want to dedicate a certain timeframe to reception call the receive function passing the maximum reception time in microseconds:

int response = bus.receive(1000);









          

      

      

    

  

    
      
          
            
  
Device ID Setup

  PJON<> bus; 
  // PJON device with NOT_ASSIGNED id in a local mode

  PJON<> bus(44); 
  // PJON device with id 44 in a local mode

  uint8_t bus_id[4] = {1, 2, 3, 4};
  PJON<> bus(bus_id, 44); 
  // PJON device id 44 in bus id 1.2.3.4 in a shared mode





Device id can also be set afterwards through a setter:

  bus.set_id(44);  





Can be read through a getter:

  bus.device_id();  









          

      

      

    

  

    
      
          
            
  
Error Handling

PJON is designed to inform the user if an error is detected. A void function has to be defined as the error handler, it receives 2 parameters the first is the error code and the second is 1 byte additional data related to the error.

Error types:


	CONNECTION_LOST (value 101), data parameter contains lost device’s id.

	PACKETS_BUFFER_FULL (value 102), data parameter contains buffer length.

	CONTENT_TOO_LONG (value 104), data parameter contains content length.



void error_handler(uint8_t code, uint8_t data) {
  if(code == CONNECTION_LOST) {
    Serial.print("Connection with device ID ");
    Serial.print(data);
    Serial.println(" is lost.");
  }
  if(code == PACKETS_BUFFER_FULL) {
    Serial.print("Packet buffer is full, has now a length of ");
    Serial.println(data, DEC);
    Serial.println("Possible wrong bus configuration!");
    Serial.println("higher MAX_PACKETS in PJON.h if necessary.");
  }
  if(code == CONTENT_TOO_LONG) {
    Serial.print("Content is too long, length: ");
    Serial.println(data);
  }
}





Now inform the bus to call the error handler function in case of error:

bus.set_error(error_handler);









          

      

      

    

  

    
      
          
            
  
Oversampling

Media: Radio, Wire |
Pins used: 1 / 2

Oversampling strategy comes from the PJON_ASK [https://github.com/gioblu/PJON_ASK] repository, and it was integrated in the PJON repository from version 3.0 beta, as a physical layer strategy. Bits are over-sampled to have high resilience in high interference scenarios like ASK/FSK cheap radio transceivers in urban environment. It is tested effectively with many versions of the ASK/FSK 315/433Mhz modules available on the market.

####Byte transmission
Every byte is prepended with 2 synchronization padding bits and transmission occurs LSB-first. The first is a longer than standard logic 1 followed by a standard logic 0. The reception tecnique is based on finding a logic 1 as long as the first padding bit within a certain threshold, synchronizing to its falling edge and checking if it is followed by a logic 0. If this pattern is recognised, reception starts, if not, interference, synchronization loss or simply absence of communication is detected at byte level.

 __________ ___________________________
| SyncPad  | Byte                      |
|______    |___       ___     _____    |
|      |   |   |     |   |   |     |   |
|    1 | 0 | 1 | 0 0 | 1 | 0 | 1 1 | 0 |
|______|___|___|_____|___|___|_____|___|





This adds a certain overhead to information but reduces the need of precise time tuning because synchronization is renewed every byte.

####Compatibility


	ATmega88/168/328 16Mhz (Diecimila, Duemilanove, Uno, Nano, Mini, Lillypad)

	ATmega2560 16Mhz (Arduino Mega)

	ATmega16u4/32u4 16Mhz (Arduino Leonardo)



####Performance


	Transfer speed: 202 B/s or 1620 Baud

	Data throughput: 150 B/s

	Range: 250 meters in urban environment / 5km with LOS



####Why not VirtualWire / RadioHead / Manchester?
I don’t think those libraries are clear, efficient and understandable enough to be the standard library for wireless radio communication available to the community, because of its implementation mess and complexity. Moreover, RadioHead doesn’t have the support for multiple devices in multimaster setup, CRC, acknowledge, collision avoidance and packet management. For this reason I wrote this implementation to provide the user with the PJON standard also on wireless. :)

####How to use OverSampling
Pass the OverSampling type as PJON template parameter to instantiate a PJON object ready to communicate in this Strategy. All the other necessary information is present in the general Documentation.

  PJON<OverSampling> bus; // 2 pin over-sampled physical layer





####Use OverSampling with cheap 433Mhz transceivers
To build a real open-source PJON packet radio able to communicate up to 5km you need only a couple (for SIMPLEX mode) or two couples (for HALF_DUPLEX mode) of cheap 315/433Mhz ASK/FSK transmitter / receiver modules (the total cost should be around 2/3 dollars). Please be sure of the regulations your government imposes on radio transmission over these frequencies before use.

[image: PJON Oversampling packet radio]

The maximum detected range was experimented with a small arduino home made packet radio transmitting its position every minute. The maximum range obtained was slightly more than 5 kilometers. Two couples of STX882 and SRX882 were used as transceivers. If you choose these modules, remember to set HIGH the pin CS on the receiver before starting reception.

Using OverSampling physical layer, synchronous acknowledge can reduce the maximum range, on certain media, so if you detect reduced range performance in HALF_DUPLEX compared to a mono-directional or SIMPLEX communication, and you can do without ACK, configure the absence of it after the packet transmission:

  bus.set_acknowledge(false);





####Antenna design
Experiments in HALF_DUPLEX mode have shown that it seems better to keep isolated the two antennas, using two different, not connected elements to transmit and receive. The first suggested antenna design is a wide beam pseudo half-wavelength dipole antenna made by two 345mm long conductive elements, one connected to ground and the other connected to the input or output pin:

      345mm                    345mm   
 -------------------|--------------------   
                  __|__                                 
                 |tx/rx|                                  
                 |_____|                                





A more directional, compact and long range antenna design is the pseudo half wavelength wip antenna. Can be easily crafted with two 345mm long insulated wire sections wrapped with each other every 5mm, one is connected to ground and the other to the input or output pin. This design helps because of its strong ground plane, often necessary to have decent results with these modules.





          

      

      

    

  

    
      
          
            
  
PJON Wiki Home

PJON (Padded Jittering Operative Network) is an Arduino compatible, multi-master, multi-media communications bus system. It proposes a Standard and it is designed as a framework to ease digital communication. Thanks to a data link layer agnostic approach the PJON protocol layer is able to communicate data regardless of the medium and the procedure used thanks to the Strategies. Its more common applications are in the field of internet of things and embedded systems. Extended tests proved its effectiveness on different media like electricity, radio and light. See the video introduction [https://www.youtube.com/watch?v=vjc4ZF5own8] to have a brief overview.

####PJON (Padded Jittering Operative Network) Protocol specification


	PJON v0.1 [https://github.com/gioblu/PJON/blob/master/specification/PJON-protocol-specification-v0.1.md]  - v0.2 [https://github.com/gioblu/PJON/blob/master/specification/PJON-protocol-specification-v0.2.md] - v0.3 [https://github.com/gioblu/PJON/blob/master/specification/PJON-protocol-specification-v0.3.md] - v1.0 [https://github.com/gioblu/PJON/blob/master/specification/PJON-protocol-specification-v1.0.md]

	PJON Acknowledge v0.1 [https://github.com/gioblu/PJON/blob/master/specification/PJON-protocol-acknowledge-specification-v0.1.md]

	PJON Dynamic addressing v0.1 [https://github.com/gioblu/PJON/blob/master/specification/PJON-dynamic-addressing-specification-v0.1.md]



####PJDL (Padded Jittering Data Link) specification


	PJDL v0.1 [https://github.com/gioblu/PJON/blob/master/strategies/SoftwareBitBang/specification/padded-jittering-protocol-specification-v0.1.md] - v1.0 [https://github.com/gioblu/PJON/blob/master/strategies/SoftwareBitBang/specification/PJDL-specification-v1.0.md]

	PJDLR v1.0 [https://github.com/gioblu/PJON/blob/master/strategies/OverSampling/specification/PJDLR-specification-v1.0.md]



####Compliant tools


	saleae-pjon-protocol-analyzer [https://github.com/aperepel/saleae-pjon-protocol-analyzer] by Andrew Grande

	PJON-python [https://github.com/Girgitt/PJON-python] by Zbigniew Zasieczny

	ModuleInterface [https://github.com/fredilarsen/ModuleInterface] by Fred Larsen




Why PJON?

PJON is a tool created to simplify communication between devices and network engineering. Choose the medium you prefer, build your own network of devices and make it work with few lines of code. There are 6 strategies available to communicate data with PJON on various media:

AnalogSampling [https://github.com/gioblu/PJON/tree/master/strategies/AnalogSampling] | Medium: Light |
Pins used: 1 (analog) or 2 (analog + digital)

AnalogSampling is designed to sample digital data using analog readings. It can be used to communicate data wirelessly through light. It is optimized to use a single LED for both data transmission and reception, enabling bidirectional half-duplex communication with a single pair of LEDs (visible light, infrared or ultraviolet) or with any sort of tuned photo-emitter and photo-receiver pair.

EthernetTCP [https://github.com/gioblu/PJON/tree/master/strategies/EthernetTCP] | Medium: Ethernet port, wired or WiFi

With the EthernetTCP PJON strategy, multiple devices with Ethernet ports can use PJON to communicate with each other on a LAN, WAN or across the Internet.

LocalUDP [https://github.com/gioblu/PJON/tree/master/strategies/LocalUDP] | Medium: Ethernet port, wired or WiFi

With the LocalUDP PJON strategy, multiple devices with Ethernet ports can use PJON to communicate with each other on a local subnet, wired or over WiFi or both.

OverSampling [https://github.com/gioblu/PJON/tree/master/strategies/OverSampling] | Medium: Radio, Wire |
Pins used: 1 or 2

Oversampling strategy comes from the PJON_ASK [https://github.com/gioblu/PJON_ASK] repository, and it was integrated in the PJON repository from version 3.0 beta, as a data link layer strategy. Bits are over-sampled to have high resilience in high interference scenarios, like using an ASK/FSK cheap radio transceivers in an urban environment. It is tested effectively with many versions of the ASK/FSK 315/433Mhz modules available on the market with up to 5km range, but it works nominally also through wires and the human body.

SoftwareBitBang [https://github.com/gioblu/PJON/tree/master/strategies/SoftwareBitBang] | Medium: Wire | Pins used: 1 or 2

SoftwareBitBang is the default data link layer strategy used by the PJON template object. This implementation is based on micros() and delayMicroseconds(). It makes no use of dedicated timers or interrupt driven strategies to handle communication. It is designed to have a small memory footprint and to be extremely resilient to interference and timing inaccuracies. Thanks to the use of a dedicated digitalWriteFast library, can be achieved fast and reliable cross-architecture communication through one or two pins.

ThroughSerial [https://github.com/gioblu/PJON/tree/master/strategies/ThroughSerial] | Medium: Hardware Serial port |
Pins used: 2

With ThroughSerial data link layer strategy, PJON can run through a software emulated or hardware Serial port. Thanks to this choice it is possible to leverage of virtually all the arduino compatible serial transceivers, like RS485, radio or infrared modules, still having PJON unchanged on top.




Practical test

After selecting the medium to be used you can start to build your personal bus (that can be made by only two devices for testing):

 Arduino UNO        Arduino UNO
  _________   wire   _________
 | ||   |_||___     |         |
 ||       ||   |    ||       ||
 ||       ||   |    ||       ||
 ||       ||   |____||       ||
 |_________|        |_|_|__||_|
                      





A simple entry level test can be to setup a couple of Arduino Duemilanove / Uno boards connected together with one single wire on both pins 12 as described in the illustration above. When the devices are wired it is possible to test their connectivity sending a packet from device 1 to device 2 and see if device 2 receives it blinking a LED:

#include <PJON.h>
PJON<SoftwareBitBang> bus(1); // <Strategy name> bus(selected device id)

void setup() {
  bus.strategy.set_pin(12);
  bus.begin();
  bus.send_repeatedly(2, "B", 1, 1000000); // Send B to device 44 every second
}

void loop() {
  bus.update();
};





As you can see the code above, device 1 is simply sending a “B” every second to the device id 2.

#include <PJON.h>
PJON<SoftwareBitBang> bus(2); // <Strategy name> bus(selected device id)

void setup() {
  pinModeFast(13, OUTPUT);
  digitalWriteFast(13, LOW); // Initialize LED 13 to be off
  bus.strategy.set_pin(12);
  bus.begin();
  bus.set_receiver(receiver_function);
};

void receiver_function(
  uint8_t *payload, 
  uint8_t length, 
  const PJON_Packet_Info &packet_info
) {
  if(payload[0] == 'B') {
    digitalWrite(13, HIGH);
    delay(30);
    digitalWrite(13, LOW);
  }
}

void loop() {
  bus.receive(1000);
};





On the receiver side is declared a receiver function that is called when a packet for the device is received. In this case, the function checks if the received character is “B” and if so, blinks the LED connected to pin 13. You should see the receiver device blinking every second.

